Approximations linéaires

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

La première version de ce document est parue en février 2006. L'objectif principal de cette feuille Maple est d'illustrer l'approximation linéaire d'une fonction d'une variable par la tangente et l'approximation linéaire d'une fonction de deux variables par le plan tangent.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.2

Initialisation

```
> restart;
> with(plots,contourplot,contourplot3d,display,setoptions,setoptions3d,
  spacecurve,textplot):
  with(plottools,point):
> setoptions(size=[300,300],labels=[x,y],tickmarks=[12,12],font=[TIMES,
  ROMAN,8],axesfont=[TIMES,ROMAN,8],labelfont=[TIMES,ROMAN,8]);
  setoptions3d(size=[300,300],labels=[x,y,z],tickmarks=[4,4,4],
  lightmodel=none,style=patchnogrid,axes=frame,axesfont=[TIMES,ROMAN,
  8],labelfont=[TIMES,ROMAN,8]);
```

L'initialisation suivante permettra d'avoir plus de lisibilité des nombres décimaux en supprimamt les zéros non significatifs à la fin d'un nombre.

```
> interface(typesetting=extended); # Pour s'assurer le niveau de
  composition étendue
  Typesetting:-Settings(striptrailing=true);
```

$$extended$$

$$false \tag{1.1}$$

```
> Digits:=20;
```

$$Digits := 20 \tag{1.2}$$

Cas d'une fonction d'une seule variable

L'équation de la tangente à la courbe d'équation $y = \mathrm{f}(x)$ au point d'abscisse $x_0 = a$, soit au point $(a, \mathrm{f}(a))$, est

$$y = \mathrm{f}(a) + \mathrm{f}'(a)\,(x - a).$$

```
> Éq_générale:=y=f(a)+D[1](f)(a)*(x-a);
```

$$Éq\_générale := y = f(a) + \mathrm{D}(f)(a)\,(x - a) \tag{2.1}$$

Soit la fonction f définie par $y = \mathrm{f}(x) = 2 + 2x^2$. La fonction f étant une fonction polynomiale, elle est donc dérivable partout. Considérons la tangente à cette courbe en $x_0 = 0.6$.

```
> f:=x->2+2*x^3;
  a:=0.6;
```

$$f := x \mapsto 2 + 2 \cdot x^3$$

$$a := 0.6 \quad \textbf{(2.2)}$$

```
> Équation_tangente:=sort(Éq_générale,x);
```

$$\text{Équation\_tangente} := y = 2.16\, x + 1.136 \quad \textbf{(2.3)}$$

Illustrons dans un même graphique la courbe f ainsi que cette tangente.

```
> plot([[x,f(x),x=0..2],[x,rhs(Équation_tangente),x=0..2]],color=[navy,
  orange],view=[0..2,0..10]);
```



Dans un même graphique, superposons les éléments que vont alimenter la discussion.

```
> p1:=plot([x,f(x),x=0..2],color=navy,numpoints=300):
  p2:=plot([x,rhs(Équation_tangente),x=0..1.8],color=orange):
  p3:=plot({[[0.6,0],[0.6,f(0.6)]],
            [[0.6,f(0.6)],[2.2,f(0.6)]],
            [[1.6,0],[1.6,f(1.6)]],
            [[1.6,f(1.6)],[2.2,f(1.6)]],
            [[1.6,eval(rhs(Équation_tangente),x=1.6)],[1.8,eval(rhs
  (Équation_tangente),x=1.6)]],
            [[1.7,f(0.6)],[1.7,eval(rhs(Équation_tangente),x=1.6)]],
            [[2,f(0.6)],[2,f(1.6)]]},
            color="Niagara 12",linestyle=2):
  p4:=textplot({[0.6,f(0.6)+.2,"( 0,6, f(0,6) )"],
                [1.6,f(1.6)+.2,"( 0,6 + &Delta;x, f( 0,6 + &Delta;x ) )
  "]},
                color=navy,align={BELOW,LEFT},font=[TIMES,ROMAN,10]):
  p5:=textplot({[1.72,3.8,"dy = f'(0,6) &Delta;x"]},
                color=orange,align={ABOVE,RIGHT},font=[TIMES,ROMAN,10])
  :
  p6:=textplot({[2.1,7,"y = f(0,6 + &Delta;x) - f(0,6)"]},
                color=navy,align={ABOVE,RIGHT},font=[TIMES,ROMAN,10]):
  p7:=textplot({[1.07,2.05,"&Delta;x"]},color=orange,font=[TIMES,
  ITALIC,10]):

  Graphique:=display([p||(1..7)],view=[0..3,0..12]):
```

```
Graphique;
```



Nous constatons que lorsque $\Delta x \to 0$, $\left(\Delta y - dy\right)$ tend vers 0 également. En posant $\Delta x = dx$, nous pouvons écrire $\left[\frac{\Delta y}{\Delta x} - \frac{dy}{dx}\right] \to 0$. Autrement dit,

$$\text{si } \Delta x \to 0, \text{ nous avons } m_{\text{sec}} - m_{\text{tan}} \to 0$$

En posant $\varepsilon = m_{\text{sec}} - m_{\text{tan}}$, on peut écrire

$$\varepsilon = \frac{\Delta y}{\Delta x} - f'(a)$$

En résumé, si f est une fonction dérivable en $x = a$, nous avons

$$\Delta y = \mathrm{f}'(a)\,\Delta x + \varepsilon \Delta x \quad (\text{où } \varepsilon \to 0 \text{ lorsque } \Delta x \to 0)$$

Puisque $\Delta y = f\left(a + \Delta x\right) - f(a) = f'(a)\,\Delta x + \epsilon \Delta x$, pour x près de a

$$\mathrm{f}\left(a + \Delta x\right) = \mathrm{f}(a) + \mathrm{f}'(a)\,\Delta x + \varepsilon \Delta x$$

Réécrivons cette égalité en posant $\Delta x = x - a$.

$$\mathrm{f}(x) = \mathrm{f}(a) + \mathrm{f}'(a)\,(x - a) + \varepsilon\,(x - a)$$

Définissons la fonction L par

$$L(x) = \mathrm{f}(a) + \mathrm{f}'(a)\,(x - a)$$

La fonction L définie $L(x) = \mathrm{f}(a) + \mathrm{f}'(a)\,(x - a)$ est appelée la linéarisation de f en $x = a$.

Nous avons, pour x près de a,

$$\mathrm{f}(x) \approx L(x)$$

On peut alors utiliser la fonction L pour approcher la fonction f dans un voisinage de $x = a$.

Résumons: si la fonction f est dérivable en $x = a$, la tangente modélise bien la fonction f au voisinage de $x = a$.

Illustrons ce fait avec des zooms avants successifs sur la courbe f et la tangente dans un voisinage de $x = 0, 6$.

```
> Graphiques:=seq(display([Graphique],axesfont=[TIMES,ROMAN,6],
                 view=[a-0.4/k..a+0.4/k,2.72-f(a-0.4/k)/k..2.72+f
  (a+0.4/k)/k]),
             k=[1,2,4,8,16,32,64,128]):
> Graphiques:=seq(plot([[x,f(x),x=a-0.2/k..a+0.2/k],[x,rhs
  (Équation_tangente),x=a-0.2/k..a+0.2/k]],color=[orange,navy],
  axesfont=[TIMES,ROMAN,6]),k=[1,2,4,8,16,32,64,128,256]):
> display(matrix(1,3,[Graphiques[1],Graphiques[2],Graphiques[3]]));
```

```
display(matrix(1,3,[Graphiques[4],Graphiques[5],Graphiques[6]]));
```





```
> a:='a':
```

Par exemple, obtenons une approximation "linéaire" de $\sqrt{17}$ au voisinage de $x = 16$.

La linéarisation L de la fonction f dérivable en $x = a$ est $L(x) = f(a) + \mathrm{D}(\mathrm{f})(a)(x - a)$.

```
> L:=x->f(a)+D(f)(a)*(x-a);
```

$$L := x \mapsto f(a) + \mathrm{D}(f)(a) \cdot (x - a) \tag{2.4}$$

Considérons la fonction f définie par $\mathrm{f}(x) = \sqrt{x}$ en $x = 16$.

```
> f:=x->sqrt(x);
  a:=16;
```

$$f := x \mapsto \sqrt{x}$$

$$a := 16 \tag{2.5}$$

La linéarisation L en $x = a$ est alors

```
> 'L'(x)=L(x);
```

$$L(x) = 4 + \frac{\sqrt{16}\,(x - 16)}{32} \tag{2.6}$$

Approximons $\sqrt{17}$ par la $L(17)$.

```
> 'L'(17)=L(17);
```

$$L(17) = 4 + \frac{\sqrt{16}}{32} \tag{2.7}$$

Comparons cette approximation avec la valeur exacte de $\sqrt{17}$.

```
> 'L(17)-sqrt(17)'=L(17)-sqrt(17);
  ``=evalf(rhs(%));
```

$$L(17)-\sqrt{17}=4+\frac{\sqrt{16}}{32}-\sqrt{17}$$

$$=0.0018943744 \qquad \textbf{(2.8)}$$

Observons la qualité de l'approximation lorsque x est davantage près de 16.

```
> 'L(16.5)-sqrt(16.5)'=evalf(L(16.5)-sqrt(16.5));
  'L(16.25)-sqrt(16.25)'=evalf(L(16.25)-sqrt(16.25));
  'L(16.125)-sqrt(16.125)'=evalf(L(16.125)-sqrt(16.125));
  'L(16.0625)-sqrt(16.0625)'=evalf(L(16.0625)-sqrt(16.0625));
```

$$L(16.5)-\sqrt{16.5}=0.0004807977$$

$$L(16.25)-\sqrt{16.25}=0.0001211259$$

$$L(16.125)-\sqrt{16.125}=0.0000303989$$

$$L(16.0625)-\sqrt{16.0625}=7.6145296507\,10^{-6} \qquad \textbf{(2.9)}$$

```
> unassign('a,f');
```

Cas d'une fonction de deux variables

L'équation du plan tangent à une surface d'équation $z=\mathrm{f}(x,y)$ au point d'abscisse $x=a$ et d'ordonnée $y=b$, soit au point $(a,b,\mathrm{f}(a,b))$ est

$$z=\mathrm{f}(a,b)+\mathrm{f}'_x(a,b)(x-a)+\mathrm{f}'_y(a,b)(y-b)$$

```
> Éq_plan_générale:=z=f(a,b)+D[1](f)(a,b)*(x-a)+D[2](f)(a,b)*(y-b);
```

$$\mathit{Éq\_plan\_générale} := z=f(a,b)+\mathrm{D}_1(f)(a,b)(x-a)+\mathrm{D}_2(f)(a,b)(y-b) \qquad \textbf{(3.1)}$$

Soit la fonction f définie par $\mathrm{f}(x,y)=-2x^2-y^2$. Cette fonction a été utilisée dans la feuille Maple intitulée « Plans tangents…un début ».

Considérons le plan tangent à cette surface en $(1,1)$.

```
> f:=(x,y)->-2*x^2-y^2:
  a:=1:
  b:=1:
  c:=f(a,b):
```

L'équation du plan tangent à la surface z au point $(1,1,\mathrm{f}(1,1))=(1,1,-3)$ est

```
> Éq_plan_tangent:=sort(Éq_plan_générale);
```

$$\mathit{Éq\_plan\_tangent} := z=-4x-2y+3 \qquad \textbf{(3.2)}$$

Illustrons, dans un même graphique, la surface f ainsi que le plan tangent au point $(1,1,\mathrm{f}(1,1))$.

```
> Surface:=plot3d([x,y,f(x,y)],x=-1..2,y=-1..2,
          grid=[30,30]):
  P:=point([1,1,f(1,1)],color="Niagara 1",
  symbol=solidsphere,symbolsize=10 ):
  Plan_tangent:=plot3d([x,y,rhs(Éq_plan_tangent)],x=-1..2,y=-1..2,
```

```
color=plum):
display(Surface,P,Plan_tangent,orientation=[95, 95]);
```



Comme l'a montré le document « Plans tangents…un début », plus nous zoomons vers l'avant, plus cela permet d'observer que la surface ressemble au plan tangent.

```
> Surface:=plot3d([x,y,f(x,y)],x=0.5..1.5,y=0.5..1.5,
        grid=[30,30]):
  Plan:=plot3d([x,y,rhs(Éq_plan_tangent)],x=0.5..1.5,y=0.5..1.5,color=
  plum):
  display(Surface,Plan,P,orientation=[-115,80,180]);
```




```
> Surface:=plot3d([x,y,f(x,y)],x=0.85..1.15,y=0.85..1.15,
        grid=[30,30]):
  Plan:=plot3d([x,y,rhs(Éq_plan_tangent)],x=0.85..1.15,y=0.85..1.15,
  color=plum):
  display(Surface,Plan,P,orientation=[-115,80,180]);
```



En définissant la fonction du premier degré L par

$$L(x, y) = \mathrm{f}(a, b) + \mathrm{D}_1(\mathrm{f})(a, b)(x-a) + \mathrm{D}_2(\mathrm{f})(a, b)(y-b)$$

cette fonction donnera de bonnes approximations de $\mathrm{f}(x, y)$ pour (x, y) proche de (a, b). La fonction L est appelée la linéarisation de f en (a, b).

```
> L:=(x,y)->f(a,b)+D[1](f)(a,b)*(x-a)+D[2](f)(a,b)*(y-b);
```

$$L := (x, y) \mapsto f(a, b) + \mathrm{D}_1(f)(a, b)\cdot(x-a) + \mathrm{D}_2(f)(a, b)\cdot(y-b) \quad \textbf{(3.3)}$$

et l'approximation

$$\mathrm{f}(x, y) \approx L(x, y)$$

est appelée l'*approximation linéaire* ou *approximation par le plan tangent* de f en (a, b).

Comparons l'approximation linéaire et la valeur de la fonction f en (1,1 ; 0,95).

Établissons d'abord la linéarisation de la fonction f en (1,1).

```
> a:=1:
  b:=1:
  'L'(x,y)=sort(L(x,y));
```

$$L(x, y) = -4x - 2y + 3 \quad \textbf{(3.4)}$$

Approximons par L.

```
> 'L'(1.1,0.95)=L(1.1, 0.95);
```

$$L(1.1, 0.95) = -3.3 \quad \textbf{(3.5)}$$

Comparons cette approximation avec la valeur exacte f(1,1 ; 0,95).

```
> 'f'(1.1,0.95)-'L'(1.1,0.95)=f(1.1,0.95)-L(1.1,0.95);
```

$$f(1.1, 0.95) - L(1.1, 0.95) = -0.0225 \quad \textbf{(3.6)}$$

L'écart de -0.0225 montre que l'approximation est "assez bonne" mais, en un point plus éloigné de (1,1), l'approximation est probablement pas aussi bonne. Comparons l'approximation et la valeur de la fonction en (2,3).

```
> 'f'(2,3)-'L'(2,3)=f(2,3)-L(2,3);
```

$$f(2, 3) - L(2, 3) = -6 \quad \textbf{(3.7)}$$

La qualité ne l'approximation est plutôt médiocre. Rappelons que l'approximation linéaire L a été établit en (1, 1).

Résumons. En posant $x = a$, $y = b$ et $z = \mathrm{f}(a, b)$, l'équation du plan tangent est

$$\mathrm{z} = \mathrm{f}(\mathrm{a}, \mathrm{b}) + \mathrm{f}'_x(\mathrm{a}, \mathrm{b})(\mathrm{x}-\mathrm{a}) + \mathrm{f}'_y(\mathrm{a}, \mathrm{b})(\mathrm{y}-\mathrm{b})$$

La fonction L définie par

$$L(x, y) = \mathrm{f}(a, b) + \mathrm{f}'_x(a, b)(x-a) + \mathrm{f}'_y(a, b)(y-b)$$

est appelée la linéarisation de f en (a, b) et l'approximation

$$\mathrm{f}(x, y) \approx \mathrm{f}(a, b) + \mathrm{f}'_x(a, b)(x-a) + \mathrm{f}'_y(a, b)(y-b)$$

est appelée ***approximation linéaire*** ou ***approximation par le plan tangent*** de f en (a, b).

IMPORTANTE REMARQUE: L'existence seules des dérivées partielles en (a, b) **n'est pas une condition suffisante** pour avoir une "bonne" approximation par le plan tangent. Considérons la fonction suivante pour faire comprendre ce fait.

Soit la fonction f définie par $\mathrm{f}(x, y) = \dfrac{x y}{x^2 + y^2}$ si $(x, y) \neq (0, 0)$ et $\mathrm{f}(0, 0) = 0$.

```
> f:=(x,y)->piecewise(x<>0 and y<>0,x*y/(x^2+y^2),x=0 and y=0,0);
```

$$f := (x, y) \mapsto \begin{cases} \dfrac{x \cdot y}{y^2 + x^2} & x \neq 0 \textbf{ and } y \neq 0 \\ 0 & x = 0 \textbf{ and } y = 0 \end{cases} \tag{3.8}$$

Traçons le graphique de cette fonction.

```
> plot3d([x,y,f(x,y)],x=-1..1,y=-1..1,grid=[60,60],orientation=[100,82,
  0]);
```



Améliorons le tracé de cette surface.

```
> a:=0.05;
  local Grid:=grid=[100,100];
  local Style:=style=patch;
```

$$a := 0.05$$
$$Grid := grid = [100, 100]$$
$$Style := style = patch \tag{3.9}$$

```
> Surface1:=plot3d([x,y,f(x,y)],x=0..a,y=0..a,Grid):
  Surface2:=plot3d([x,y,f(x,y)],x=-a..0,y=0..a,Grid):
  Surface3:=plot3d([x,y,f(x,y)],x=-a..0,y=-a..0,Grid,Style):
  Surface4:=plot3d([x,y,f(x,y)],x=0..a,y=-a..0,Grid,Style):
> Vue:=display(Surface||($1..4),orientation=[100,82,0]):
  Vue;
```



Bien que la fonction f soit définie en (0,0), elle n'est pas continue en (0,0). En effet, la limite en $(0, 0)$ n'existe pas.

Obtenons la limite par le chemin $y=x$.

```
> limit(subs(y=x,f(x,y)),{x=0,y=0});
```

$$\frac{1}{2} \tag{3.10}$$

Obtenons ensuite la limite par le chemin $y=-x$.

```
> limit(subs(y=-x,f(x,y)),{x=0,y=0});
```

$$-\frac{1}{2} \tag{3.11}$$

Ce qui montre que la limite lorsque $(x, y) \to (0, 0)$ n'existe pas. Montrons maintenant que les dérivées partielles existent en (0,0). Calculons chaque dérivée partielle à partir de la définition.

```
> `f '`[x](0,0)=Limit(('f'(0+h,0)-'f'(0,0))/h,h=0);
  ``=value(rhs(%));
```

$$f'_x(0,0) = \lim_{h \to 0} \frac{f(h,0) - f(0,0)}{h}$$

$$= 0 \tag{3.12}$$

et

```
> `f '`[y](0,0)=Limit(('f'(0,0+h)-'f'(0,0))/h,h=0);
  ``=value(rhs(%));
```

$$f'_y(0,0) = \lim_{h \to 0} \frac{f(0,h) - f(0,0)}{h}$$

$$= 0 \tag{3.13}$$

Ce qui montre que les dérivées partielles à l'origine existent et valent 0, à savoir $f'_x(0,0) = 0$ et $f'_y(0,0) = 0$. L'approximation linéaire serait dans ce cas $f(x,y) \approx 0$. Or, $f(x,y) = \frac{1}{2}$ tout le long de la droite $y=x$. C'est un bel exemple phare qui nous montre que l'existence seules des dérivées partielles ne nous assurent pas que la fonction puisse être approchée avec un plan tangent.

De plus, cet exemple nous montre ceci: bien que les dérivées partielles d'une fonction f existent en (a, b), cela n'implique pas que la fonction f soit continue en (a, b).

Afin de pouvoir approcher une fonction f avec sa linéarisation L (plan tangent), nous devons recourrir à la notion de différentiabilité d'une fonction de deux variables.

Si $z = f(x,y)$, alors f est différentiable en (a, b) si Δz peut être écrit sous la forme

$$\Delta z = f'_x(a,b)\,\Delta x + f'_y(a,b)\,\Delta y + \varepsilon_1\,\Delta x + \varepsilon_2\,\Delta y \quad \text{où } (\varepsilon_1, \varepsilon_2) \to (0,0) \text{ lorsque } (\Delta x, \Delta y) \to (0,0)$$

Si cette définition est satisfaite, cela nous assure que le plan tangent modélise bien la fonction f à proximité de (a, b).

Enfin, si les dérivées partielles f'_x et f'_y **existent** à proximité de (a, b) et sont **continues** en (a, b), alors f est différentiable en (a, b). Donc, pour garantir que le plan tangent en (a, b) modélise bien la fonction en (a, b), il suffit de s'assurer de la continuité des dérivées partielles en (a, b).

Par exemple, soit la fonction f définie par $f(x,y) = x e^{xy}$. Estimons $f(1.1, -0.1)$ avec une approximation linéaire.

```
> f:=(x,y)->x*exp(x*y):
  'f'(x,y)=f(x,y);
```

$$f(x, y) = x\,e^{yx} \tag{3.14}$$

Calculons les dérivées partielles.

```
> `f '`[x](x,y)=D[1](f)(x,y);
  `f '`[y](x,y)=D[2](f)(x,y);
```

$$f'_x(x, y) = e^{yx} + x\,y\,e^{yx}$$

$$f'_y(x, y) = x^2\,e^{yx} \tag{3.15}$$

Puisque ces deux dérivées partielles sont continues (pourquoi ?), nous pouvons modéliser la fonction f par son plan tangent en (1,0). Construisons donc la linéarisation de f en (1,0).

```
> L:=(x,y)->f(a,b)+D[1](f)(a,b)*(x-a)+D[2](f)(a,b)*(y-b);
```

$$L := (x, y) \mapsto f(a, b) + D_1(f)(a, b)\cdot(x-a) + D_2(f)(a, b)\cdot(y-b) \tag{3.16}$$

```
> a:=1:
  b:=0:
```

Alors, la linéarisation de la fonction f en (1,0) est

```
> 'L'(x,y)=sort(L(x,y));
```

$$L(x, y) = x + y \tag{3.17}$$

Estimons $f(1.1, -0.1)$.

```
> 'L'(1.1,-0.1)=L(1.1,-0.1);
```

$$L(1.1, -0.1) = 1. \tag{3.18}$$

Comparons cette valeur avec la valeur exacte de la fonction.

```
> 'L'(1.1,-0.1)-'f'(1.1,-0.1)=L(1.1,-0.1)-f(1.1,-0.1);
```

$$L(1.1, -0.1) - f(1.1, -0.1) = 0.0145824512 \tag{3.19}$$

Observons la qualité des approximations suivantes au fur et à mesure que l'on s'approche de $(1, 0)$.

```
> 'L'(1.1,-0.1)-'f'(1.1,-0.1)=L(1.1,-0.1)-f(1.1,-0.1);
  'L'(1.05,-0.05)-'f'(1.05,-0.05)=L(1.05,-0.05)-f(1.05,-0.05);
  'L'(1.01,-0.01)-'f'(1.01,-0.01)=L(1.01,-0.01)-f(1.01,-0.01);
  'L'(1.001,-0.001)-'f'(1.001,-0.001)=L(1.001,-0.001)-f(1.001,-0.001);
  'L'(1.0001,-0.0001)-'f'(1.0001,-0.0001)=L(1.0001,-0.0001)-f(1.0001,
  -0.0001);
```

$$L(1.1, -0.1) - f(1.1, -0.1) = 0.0145824512$$

$$L(1.05, -0.05) - f(1.05, -0.05) = 0.0037029629$$

$$L(1.01, -0.01) - f(1.01, -0.01) = 0.0001496579$$

$$L(1.001, -0.001) - f(1.001, -0.001) = 1.499665792\;10^{-6}$$

$$L(1.0001, -0.0001) - f(1.0001, -0.0001) = 1.4999666579\;10^{-8} \tag{3.20}$$